Add-In

door Walther Musch

Inleiding

Hoe het gebruik van een Add-In het programmeren kan vereenvoudigen en uniform kan maken

VB6

Wie al jaren programmeert heeft al vaak vrijwel dezelfde code in zitten typen. Er zijn wel wat hulpmiddelen om code te hergebruiken (o.a. Copy&Paste). Maar de kans dat daarna de code alsnog moet worden aangepast blijft ook dan groot.

Handig zou zijn als je snel het raamwerk van een functie, procedure of wat maar ook kunt produceren.

Met een Add-In is dit op een simpele manier te bereiken. Op zich is veel kennis van de VB ontwikkelomgeving niet nodig. (Het kan zo ingewikkeld gemaakt worden als men wil..) Er zijn maar een paar functies nodig. Dit zijn allemaal proporties van het VBIDE.VBE object:

.ActiveCodePane.CodeModule.InsertLines

.ActiveCodePane.GetSelection

Het maken van een Add-In gaat het snelst via de templates: Open een nieuw project en kies hierbij voor ‘Add-In’. Het project heeft standaard een form (frmAddin) en een designers (Connect). Plaats op het form een tekstveld (txtInvoer). 

Voeg de volgende code toe:

Private mlngLineStart As Long

Private mlngColStart As Long

Private mlngLineEnd As Long

Private mlngColEnd As Long

Onder het click_event van de OKButton wordt de volgende code geplaatst:

Private Sub OKButton_Click()

VBInstance.ActiveCodePane.GetSelection mlngLineStart,_

 mlngColStart, mlngLineEnd, mlngColEnd

 With VBInstance.ActiveCodePane.CodeModule

  Call .InsertLines(mlngLineStart,txtInvoer.Text)

 End With

End Sub

Druk op Control-F5. Open een nieuw project; kies voor een ‘Standard EXE’. Onder het menu ‘Add-Ins’ is nu een nieuwe optie bijgekomen: ‘My AddIn’. Open het standaard form (Form1) in ‘code view’ en start dan ‘My AddIn’. Toets wat in het tekstvenster en druk dan op ‘OK’. De ingebrachte tekst wordt gekopieerd naar het openstaande VisualBasic project. In feite is hiermee het principe werkbaar gemaakt.

Natuurlijk moet er wat ‘intelligentie’ worden ingebouwd om de juiste tekst in te voegen. Stel dat de ontwikkelaar alleen maar de naam van een functie hoeft te geven en dat daarna het raamwerk van de functie klaar wordt gezet compleet met foutafhandeling.

Voeg daarvoor de volgende code toe:

Private Sub CreateFunction()

    Dim mstrLinesToInsert  As String

    '

    mstrLinesToInsert = "Public Function " + _

        txtInvoer.Text + "() AS Long" & _

        vbCrLf & vbTab & "ON ERROR GOTO " & _

        txtInvoer.Text & "_FAIL" & vbCrLf & _

        vbTab & "'" & vbCrLf & txtInvoer.Text & _

        "_EXIT:" & vbCrLf & vbCrLf & _

        vbTab & "Exit Function" & vbCrLf & _

        txtInvoer.Text & "_FAIL:" & vbCrLf & _

        vbTab & "Call Err.Raise(Err.Number," & _

        Chr(34) & txtInvoer.Text & Chr(34) & _

        ", " & Chr(34) & txtInvoer.Text & Chr(34) & _

        ")" & vbCrLf & vbTab & "Resume " & _

        txtInvoer.Text & "_EXIT" & vbCrLf & _

        "End Function"

    Call TransferSelection(mstrLinesToInsert)

End Sub

Private Sub TransferSelection(ByRef strText As String)

   VBInstance.ActiveCodePane.GetSelection _ 

     mlngLineStart, mlngColStart, _

     mlngLineEnd, mlngColEnd

    With VBInstance.ActiveCodePane.CodeModule

        Call .InsertLines(mlngLineStart,strText)

    End With

End Sub

Wijzig de code onder het click_event van de OKButton:

Private Sub OKButton_Click()

    Call CreateFunction

End Sub

Compileer het project. Start een nieuw standaard executabel project en geef de naam van een functie die toegevoegd moet worden.

Dit voorbeeld is te statisch om in bruikbaar te zijn in de praktijk. Er wordt niet alleen gebruik gemaakt van functies met als resultaat een Long. Er zijn tientallen varianten te bedenken. En dan hebben we het nog niet over Procedures, Proporties en eventuele andere veel gebruikte code. Het scherm zal dan uitgebreid moeten worden met allerlei opties. 

· Er is ook een proportie die aangeeft wat voor soort Visual Basic CodePanel (module, class, form etc) openstaat. Deze kan gebruikt worden om bijvoorbeeld in de foutafhandeling behalve de functienaam ook de modulenaam mee te geven. Deze naam is op te halen met:

VBInstance.ActiveCodePane.CodeModule.Parent.Name

De foutafhandeling kan dan worden aangepast met

mstrContainerName= VBInstance.ActiveCodePane.CodeModule.Parent.Name

"'Call Err.Raise(Err.Number, " & Chr(34) & _

mstrContainerName & Chr(34) & ", " & _

Chr(34) & txtName.Text & Chr(34) & ")" & vbCrLf

· Het type van de openstaande CodePanel kan worden opgevraagd met:

VBInstance.ActiveCodePane.CodeModule.Parent.Type

Aan de hand van het resultaat kan dan weer het een en ander wel of niet getoond worden. Bijvoorbeeld op een module mag geen Proportie van het type Friend worden toegevoegd. Deze kan dan in code omgezet worden naar het type Private.

De belangrijkste typen zijn:

1 = module

2 = class

5 = form

8 = usercontrol

· Door het gebruik van de functie GetSelection wordt de code ingevoerd op de plek waar de cursor stond. Het kan ook zijn dat het noodzakelijk is om een gebruikte variabele toe te voegen aan de ‘Declarations’. Om dit te bewerkstellingen is er de proportie VBInstance.ActiveCodePane.CodeModule.CountOfDeclarationLines. Deze geeft het aantal regels terug in de sectie. Om tekst hieraan toe te voegen kan de volgende code worden gebruikt:

With VBInstance.ActiveCodePane.CodeModule

 Call .InsertLines(VBInstance. _   

  ActiveCodePane.CodeModule._

  CountOfDeclarationLines + 1, strText)

End With

Kortom met slechts vijf verschillende functies en het standaard template project kan er al snel een werkbare AddIn worden gemaakt. Er moet nog even worden uitgedokterd hoe de code er precies uit moeten komen te zien en misschien moet er een duidelijke naam voor de AddIn nog worden gezet maar dan is het ook klaar. Het voordeel van deze manier van werken is dat de code er vrij uniform uitziet. Dat is altijd meegenomen bij het onderhoud en versiebeheer. Verder kunnen bepaalde afspraken binnen een ontwikkelteam worden afgedwongen (bijvoorbeeld het gebruik van foutafhandeling).

Het project (in code; vrij te gebruiken) waaraan de schermafbeelding is ontleend is te vinden op o.a. http://www.kather.net/VisualBasicSource/Publish
